
Рассмотрим способ получения 8-байтового истинно уникального идентификатора, пригодного для использования в качестве первичного ключа записи. Такой идентификатор GUID ( Globally Unique Identifier ) является основой для системного построения таких механизмов, как ссылочные отношения между таблицами, выделение (подсвечивание) записей таблицы, семафорные блокировки объектов, быстрые алгоритмы на основе SQL. Для получения такого идентификатора мы должны решить следующие задачи: 

· Получение идентификатора компьютера 

· Получение идентификатора программной сессии

· Получение идентификатора уникального значения таймерного счетчика

· Оптимизация быстродействия (ожидания) значения таймерного счетчика

· Компактная упаковка полученных идентификаторов в единое значение GUID

· Возможность восстановления даты и времени создания из GUID

Рассмотрим все перечисленные задачи в виде частных функций. Предварительно необходимо решить задачу представления десятичных чисел в виде компактных чисел в системе исчисления с большим основанием, в нашем случае - с основанием 150. Эта цифра получается после отбраковки всех неподходящих символов из набора ASCII. Именно с помощью ASCII символов мы будем представлять число с большим основанием. Набор символов представления получим с помощью функции GetUniString:

*** Построение массива символов для уникального кода

Function GetUniString

Local cUniS,ni

&& cUnis - это набор знаков для отображения чисел с основанием LEN(cInis)

&& выкинули символы: [0-34] '"[]& (39,34,91,93) 

*** набираем строку знаков

cUnis=chr(35)+chr(36)+chr(37) && +chr(38) - "&"

For ni=40 to 90

  cUniS=cUniS+chr(ni)

EndFor

   cUniS=cUniS+chr(92)

      For ni=94 to 126

        cUniS=cUniS+chr(ni)

      EndFor

         For ni=192 to 255

           cUniS=cUniS+chr(ni)

         EndFor

Return cUniS


Десятичное число представить в виде числа с другим основанием системы исчисления позволяет функция NumtoB() :

* Преобразование числа nNum в строку символов

* это фактически упаковка числа в строку символов

* строка используемых знаков находится в переменной UniString_

* (ф-я VFP BinToC() вылетает при числе разрядов > 9)

* обратное преобразование - BtoNum

*** Пример выбора систем представлени

*UniString_="01" - 2

*UniString_="0123456789" - 10

*UniString_="0123456789ABCDEF" - 16

Function NumtoB(nNum)

Local cBuf,ni,nOst,nTial,nOsn,nVes, nNum, nLen, cUniStr

cBuf=""

nOst=0

nTail=0

nLen=1

cUniStr=UniString_

*** Основание итогового числа - длина массива используемых знаков UniString

nOsn=Len(cUniStr)&&-1 && диапазон визуальных символов для строки

*** Вычислим количество разрядов в итоговом числе (ex.999>123)

DO WHILE .T.

        nOst=nOst+(nOsn-1)*nOsn**(nLen-1)

    IF nOst >= nNum

       EXIT

    ENDIF

  nLen=nLen+1

ENDDO

FOR ni=1 TO nLen

     nOst=nNum-nTale

 * вес ni разряда - каждый разряд может принимать значение

 * от старщего до младшего элемента UniString

     nVes=INT(nOst/nOsn**(nLen-ni))

 * текущий остаток от границы текущего разряда

     nTail=nTail+nVes*nOsn**(nLen-ni)

 * по весу разряда извлекаем символ из строки уникальных знаков UniStr

 * nVes+1 для получения знака по nVes==0

      cBuf=cBuf+SUBSTR(cUniStr,nVes+1,1)

NEXT

RETURN cBuf

 

Сразу же предусмотрим функцию BtoNum(), позволяющую восстановить десятичное число из упакованного:

 *** Обратное преобразование в десятичное число

Function BtoNum(cB)

Local ni,nj,nOst,nk,nOsn,cCim,cUniStr

  nOst=0

  cUniStr=UniString_

  nOsn=LEN(cUniStr)

FOR ni=1 TO LEN(cB)

   cCim=SUBSTR(cb,LEN(cb)-ni+1,1)

      nk=AT(cCim,cUniStr)-1

        nOst=nOst+nk*(nOsn**(ni-1))

NEXT

RETURN nOst


Уникальное значение таймерного счетчика получим
с помощью физической модели хронологического ряда. Выберем точку отсчета, например, с 1 января 2000 года. Это будет начальным смещением. Смещение предпочтительнее, т.к. нет смысла хранить все время с рождества Христова. От точки отсчета будем считать прошедшее число секунд. Их число мы получим как сумму миллисекунд от точки отсчета до текущей даты и числа миллисекунд, прошедших от начала текущих суток:

* Получить число тиков (10 мс - точнее не может NT) прошедшее с 2000 года 

* seconds() возвращает секунды с точностью до миллисекунд

* 6 байт - хватает на 20 лет

nStartDate_ - число дней, прошедших к 01-01-2000

Function Tic96

Return int(((val(sys(11,date())) - nStartDate_))*8640000+seconds()*100)


Получить уникальное значение счетчика мы можем только при ожидании наступления очередной миллисекунды. Такой алгоритм при генерации большого пакета идентификаторов за одну секунду сможет породить только 100 значений. Несмотря на то, что обычно идентификатор в реальной ситуации создается с меньшей частотой, предусмотрим более эффективное быстродействие. Для этого введем дополнительные разряды счетчика, которые будет инкрементироваться при нескольких вызовах GUID до наступления очередной миллисекунды. Выберем счетчик на 100 значений, что позволит делать 10000 вызовов в секунду. Для более высокой частоты генерации необходимо повысить разрядность дополнительного счетчика. При переполнении дополнительного счетчика будем дожидаться наступления следующей миллисекунды. Такая модель дает 100% достоверность уникальности числа. Алгоритм приведен в функции Tic96N:

* Получить уникальное число - тики Tic96 + счетчик-дополнение

* В пределах одного тика Tic96 может вызваться несколько раз,

* поэтому вводим дополнительно increment, old - in nd96_

Function Tic96N

Local nNew,nNewS

nNewS=Tic96()

  IF nNewS==nd96s_ && все еще старый тик

     IF nd96d_==99 && уже нет запаса счетчика - тянем новый тик

          DO WHILE nNewS==nd96s_

              nNewS=Tic96()

          ENDDO

       nd96d_=0

     ELSE

       nd96d_=nd96d_+1 && Запас есть - увеличиваем счетчик

     ENDIF 

  ELSE && новый тик

    nd96d_=0 

  ENDIF

nNew=nNewS*100+nd96d_ && Получим новый Id

nd96s_=nNewS

return (nNew)


Выбранная модель генерации и упаковки позволяет в 6 байтах сохранить значение числа 0.0001 секунд за период в 20 лет. Иначе говоря, 100% уникальность будет соблюдена даже в случае непрерывной и одновременной генерации записей с первичными ключами на нескольких изолированных системах в течение 20 лет. 

При одновременной работе нескольких изолированных друг от друга информационных систем, хоть и малая, но может существовать вероятность непрерывного, одновременного и параллельного вызова генераторов GUID. В этом случае есть ненулевая вероятность получения одинаковых значений временных счетчиков. При добавлении к значению временного счетчика идентификатора компьютера эту вероятность мы сведем к нулю. Вероятность дублирования значений после сдвига системных часов мы сведем к нулю путем добавления еще одного идентификатора - программной сессии. Заметим, что можно использовать и несколько измененный метод построения дополнительного (к временной части) кода. Увеличив емкость расчетного значения GUID на несколько дополнительных байтов, мы можем поднять надежность расчета уникального значения применительно к возможным случаям сдвига системных часов во время исполнения программной сессии. С этой целью к идентификатору компьютера добавим еще третью компоненту - код, полученный с помощью генератора случайных чисел. 

При запуске приложения генератор инициализируется, строя идентификатор компьютера и идентификатор сессии - еще 2 байта:

*** Инициализация генератора GUID - вызывать в начале приложения

Function IniGenGuid

Public hid2_guid_, UniString_, nd96s_, nd96d_, nStartDate_ && счетчики для идентификатора GUID

Local cBuf

  nStartDate_ = VAL(SYS(11,CTOD("01/01/2000"))) && дата начала отсчета счетчика

  nd96s_=0

  nd96d_=0

    *** Загрузка массива символов для уникального кода

    UniString_=GetUniString() && сохраним для быстрого извлечения

    *** инициализация генератора GUID

  cBuf=sys(0)+sys(2020) && получим id компьютера (+ используем функцию GetVolInfo от Benn Kjaer)
  cBuf=cBuf+Sys(2015) && подмешаем в аппаратный код код сессии (при сдвиге часов он обеспечит unique)

  Hid2_Guid_=PADR(NumToB(Val(sys(2007,cBuf))/3),2,'*') && CheckSum idPc+idSession - 2 bytes (max n=65000/3)

  nd96s_=Tic96() && последнее значение тиков

RETURN


Системно организованная модель генерации позволяет нам производить сортировку по первичному ключу в порядке хронологии создания ключа, восстанавливать дату и время из него:

** Восстановить дату из Guid

Function DatFromGuid(cGuidValue)

 Local nTic, dDate

  nTic=BTONum(ALLTRIM(SUBSTR(cGuidValue,1,6)))

  nTic=nTic/100 && усечем 2 доп.разряда счетчика

    dDate=INT(nTic/8640000) && дни

    dDate=CTOD(SYS(10,dDate+nStartDate_))

RETURN dDate

 
** Восстановить время из GUID 

Function TimFromGuid(cGuidValue)

Local nSec, cTim, cHour, cMin, cSec

nSec=SecFromGuid(cGuidValue)

cHour=INT(nSec/3600)

cMin=INT((nSec-cHour*3600)/60)

cSec=nSec-cHour*3600-cMin*60

cHour=PADL(ALLTRIM(STR(cHour,2,0)),2,"0")

cMin=PADL(ALLTRIM(STR(cMin,2,0)),2,"0")

cSec=PADL(ALLTRIM(STR(cSec,5,2)),5,"0")

cTim=cHour+":"+cMin+":"+cSec

Return cTim
Приведем теперь собственно функцию GUID():

* Основная функция - Получить уникальное значение GUID (8 байт) = TimeStamp  + Hard_id

Function GUID

* 6b временной счетчик + 2b hard-конфигурации+сессии

Return PADR(PADL(NumtoB(Tic96n()),6)+Hid2_guid_,8,'!')

Для второй схемы (с дополнительным использованием генератора случайных чисел) функция будет выглядеть следующим образом:

* Основная функция - Получить уникальное значение GUID (10 байт) = TimeStamp + Hard_id + Random value

Function GUID

* 6b временной счетчик + 2b hard-конфигурации/сессии + 2b динамический случайный код

Return PADR( PADL(NumtoB(Tic96n()),6) + Hid2_guid_ + PADR(NumToB(RAND()*10**4),2,'*'),10,'!')

 
Теперь все готово для вызова нашего генератора GUID. Для этого достаточно издать вызов функции GUID():

Local cBuf

=IniGenGuid() && инициализировали генратор

cBuf=GUID() && получили значение GUID

=Messagebox("Это значение GUID: "+cBuf+chr(10)+chr(13)+;

"А это дата, извлеченная из GUID: "+DTOC(DatFromGuid(cBuf)))

Return
2000г, Кубанския Госуниверситет

